
ABSTRACT

The objective of this work was to implement and evaluate a method for generating
skewed random numbers using a combination of uniform random numbers. The method
provides a simple and accurate way of generating skewed random numbers from the
specified first three moments without an a priori specification of the probability density
function. We describe the procedure for generating skewed random numbers from
uniform random numbers, and show that it accurately produces random numbers with the
desired first three moments over a range of skewness values. We also show that in the
limit of zero skewness, the distribution of random numbers is an accurate approximation
to the Gaussian probability density function. Future work will use this method to provide
skewed random numbers for a Langevin equation model for diffusion in skewed
turbulence.

1. INTRODUCTION

The motivation for this work on skewed random number generation is the study of the
dispersion of pollutants in the convective boundary layer, CBL. The CBL is the daytime,
unstably-stratified atmospheric boundary layer, and typically extends from the ground
surface to a height of 1 km. Pollutant dispersion within the CBL is qualitatively different
from that within a stably- or neutrally-stratified boundary layer, where dispersion can be
described by standard Gaussian (unskewed) turbulence models.

During the last two decades, a clear understanding has emerged of the complex processes
involved in the time-averaged vertical dispersion of a pollutant in the CBL. For an
elevated source, the height of maximum concentration of pollutant exhibits the behavior
of approaching the ground surface with increasing time and downwind distance. For a
source at the ground surface, the height of maximum concentration increases with
downwind distance. The initial understanding of these behaviors emerged from a
combination of laboratory (Willis and Deardorff; 1976, 1978, 1981) and numerical
studies (Lamb; 1978a, 1978b, 1982). Confirmation of these phenomena in the atmosphere
has been found more recently in a field study by Briggs (1993).

These behaviors cannot be explained by the first moment (mean) of the vertical wind
velocity, w , which is typically zero, nor by the second moment, w2 . However, they can
be explained by the third moment, w3 . In the CBL, w3 > 0  and correspondingly the
probability distribution of vertical velocity, P(w), is positively skewed, where skewness is
defined as S = w3 w2( )3/ 2

. Positive skewness in the vertical wind velocity is the result of
strong solar heating of the ground surface, e.g., during cloudless midday conditions. This
heating generates strong updrafts or thermals (w > 0) over approximately 40% of the
horizontal area, on average, and weak downdrafts (w < 0) over the remaining 60%.
Consequently, a plume from an elevated source has a higher probability of encountering a
downdraft, so the locus of maximum concentration decreases in height with downwind
distance. For near-surface releases, plumes travel horizontally near the ground surface
until they become incorporated in an updraft, causing an increase in height of the
maximum concentration.

The impact of these phenomena on the ground level concentration is of great importance,
especially for elevated sources. Significant underprediction, by as much as a factor of 2.9
(Briggs, 1993), can occur if these processes are not taken into account in atmospheric
dispersion models. Since accurate prediction of ground-level concentrations are critical to
atmospheric dispersion modeling applications such as environmental impact assessment,
safety analysis, and emergency response, these effects need to be included.

Atmospheric dispersion models based on the Langevin equation have been developed
which attempt to simulate turbulent dispersion in the skewed turbulence of the CBL. The

most successful of these models have used a Gaussian (unskewed) random force (e.g.,
Thomson, 1987; Luhar & Britter, 1989; and Weil, 1990). We are developing a Langevin
equation model that uses a skewed random force, but which overcomes limitations of
existing skewed-random-force models. To implement such a model, a method of
generating skewed random numbers from the specified first three moments is required.
Such a method is described and tested below.

2. THE DOUBLE-BLOCK DISTRIBUTION

We have developed a simple method which can be used to generate skewed random
numbers. It uses a combination of two overlapping uniform probability distributions,
which shall henceforth be referred to as a "double-block" distribution. Uniform
distributions are the simplest probability distributions, and uniformly-distributed random
numbers are easily generated using computer pseudo-random number generators.

Figs. 1a-b show an example of the combination of two overlapping uniform probability
density functions. Fig. 1a shows the two individual probability density functions, P1(b)
and P2(b), defined by six parameters: the means m1 and m2, half-widths ∆1 and ∆2, and
probability densities p1 and p2. These parameters have the following properties:

∆1> |m1|  and

∆2> |m2|.

Therefore, the two distributions overlap each other and both cover b = 0.

Fig. 1b shows the double-block probability density function, P(b), which is the sum of
these two uniform distributions, i.e.,

P(b)  = P1(b)  + P2(b) , (1)

where

P1(b) =
p1,  if (m1 − ∆1) < b < (m1 + ∆1)

0 ,  elsewhere




(2)

and

P2 (b) =
p2,  if (m2 − ∆2 ) < b < (m2 + ∆2 )

0 ,  elsewhere




(3)

m2m1

p1

∆2∆1

p2

P(b)

b

Figure 1a. Example of two overlapping uniform probability density functions,
P1(b) and P2(b) with means m1 = –0.51 and m2 = 0.73; half-widths ∆1=1.1 and
∆2 =1.6; and probability densities p1 = 0.26 and p2 = 0.13, respectively.

(p1+p2)

b

P(b)

Figure 1b. Example of double-block probability density function, P(b), which
is the sum of the two overlapping uniform distributions in Fig. 1a.  This
distribution has a mean of zero, a variance of 1 and a skewness of 0.5.

In general, only the first three moments of the vertical wind velocity are known.
Consequently, we will derive the six parameters of this distribution so that the desired
first moment (the mean, assumed to be zero), second moment (σ2), and third moment (ζ3)
result.  Moments zero through three provide us with four equations:

b0 = P(b)db = 1,

−∞

∞

∫

b = bP(b)db = 0,

−∞

∞

∫

b2 = b2P(b)db = σ 2 ,

−∞

∞

∫  and

b3 = b3P(b)db = ζ 3.

−∞

∞

∫

(4a)

(4b)

(4c)

(4d)

Using Eqs. (1 -3), Eqs. (4a-d)  become

2 p1∆1 + 2 p2∆2 = 1, (5a)

m1p1∆1 + m2 p2∆2 = 0, (5b)

2
3 p1∆1

3 + 2∆1m1
2 p1 + 2

3 p2∆2
3 + 2∆2m2

2 p2 = σ 2, and (5c)

2 ∆1
3m1p1 + ∆1m1

3 p1 + ∆2
3m2 p2 + ∆2m2

3 p2[ ] = ζ 3 . (5d)

Since there are four and equations and six unknowns, two more equations are required for
closure.  We use the following two equations:

∆1 = −am1  and (5e)

∆2 = am2 . (5f)

where a is a positive constant (assuming m1 < 0 and m2 > 0), which must be specified.
For a greater than one, the two distributions overlap at b = 0. We have chosen a = 5 ,
which gives the best fit to a Gaussian distribution (when ζ 3 is zero).

Assuming a = 5  and solving Eqs. (5a-f) yields

m1 = 2

9σ 2 ζ 3 − ζ 6 + 243
32 σ6( )1

2




, (6a)

m2 = 2

9σ 2 ζ 3 + ζ 6 + 243
32 σ6( )1

2




, (6b)

p1 = m2

2am1 m1 − m2( ) , and (6c)

p2 = m1

2am2 m1 − m2( ) . (6d)

Parameters ∆1 and ∆2 are then defined by eqs. (5e- f).

These equations (5e,f; 6a-d) completely define the double-block probability function
P(b), given the desired moments, b = 0 , b2 = σ 2 , and b3 = ζ 3 . A set of random numbers,
{bi}, can then be generated with these desired first three moments. An algorithm for
doing this (which was used for the sample calculations shown below in section 4) is as
follows:

Double-block random number algorithm

(1) For the desired values of σ 2 and ζ 3, calculate m1, m2, ∆1, ∆2, and p1

      using Eqs. (6a), (6b), (5e), (5f), and (6c), respectively.

(2) For each random number bi:

(a) Obtain two uniformly-distributed random
numbers ri′ and ri″ on (0,1)

(b) If ri′ < 2∆1p1 (total probability of first uniform distribution),
then

bi = m1 + 2∆1(ri
″ − 0.5)[ ]

else

bi = m2 + 2∆2 (ri
″ − 0.5)[ ]

An alternate method which requires only one uniform random number is as follows:

Alternate double-block random number algorithm

(1) For the desired values of σ 2 and ζ 3, calculate m1, m2, ∆1, ∆2, and p1

      using Eqs. (6a),  (6b), (5e), (5f), and (6c), respectively.

(2) For each random number bi:

(a) Obtain one uniformly-distributed random
numbers ri on (0,1)

(b) If ri < 2∆1p1 ≡ P1 (total probability of first uniform distribution),
then

ri′ = ri / P1

bi = m1 + 2∆1(ri
′ − 0.5)[ ]

else
ri′ = (ri – P1)/ (1 –P1)

bi = m2 + 2∆2 (ri
′ − 0.5)[ ]

3. COMBINATIONS OF DOUBLE-BLOCK DISTRIBUTIONS

While double-block distributions as in Fig. 1b, have the desired first three moments, they
are not smooth and continuous. However, skewed random numbers with continuous
distributions, and with the desired first three moments, can be generated from
combinations of double-block random variables. Such a random variable, B, can be
defined as follows

B = 1

N
1

2
bi

i=1

N

∑ , (7)

where bi is a double-block random number and N is the number of double-block random
numbers used to generate B.  The moments of B are related to the moments of b, as
follows:

 B = 1

N
1

2
bi

i=1

N

∑ = 0 , (8a)

B2 = 1

N
1

2
bi

i=1

N

∑







2

= 1
N

bi
2

i=1

N

∑ = b2  , (8b)

B3 = 1

N
1

2
bi

i=1

N

∑







3

= 1

N
3

2
bi

3

i=1

N

∑ = N−1
2 b3 . (8c)

Therefore, random numbers, B , can be generated with desired moments
B (= 0),  B2 ,  and B3  by using Eq. ( 7) with double-block random numbers, bi,  which
have the following moments:

b = 0 , (9a)

b2 = σ 2 = B2 (9b)

b3 = ζ 3 = N
1

2 B3 (9c)

Note that the third moment of b, b3 = ζ 3 , increases in a manner that is proportional to
N1/2, where N is the number of double-block random numbers, bi,  per random number B.
This is necessary in order to maintain the desired skewness of B regardless of the number
of double-block random numbers, bi, that are combined to form B. If b3  did not depend
on N, then, for large N, the Central Limit Theorem states that the distribution of B as N
becomes large will approach a Gaussian distribution, which has zero skewness.

This method was tested for several values of skewness,

S = B3

B2( )
3

2
,

and for several values of N. A FORTRAN computer code which generates a sample of
random numbers using this method, and calculates the corresponding histogram and
probability density function is listed in the Appendix. Sample calculations using this
method are given in the next section.

4. SAMPLE CALCULATIONS

Figs. 2-4 show calculated probability density values for B using the above method for
three values of skewness (S = 0.0, 0.5, and 1.5) and for five values of N (N = 1, 2, 3, 5,
and 10). All calculations are for a desired mean of 0, desired variance (σ2) of 1, and a
sample of 106 values of B. The number of occurrences of B values in evenly-spaced bins
from B = –3σ to +3σ were counted. For Figs. 2a-e, twenty four bins were used. For Figs.
3a-e and 4a-e, forty eight bins were used. Probability density values for each bin were
then calculated and plotted as dots.

Figs. 2a-e show the calculated probability density for zero skewness for values of N = 1,
2, 3, 5, and 10. Also plotted (as a solid line) in these figures is the corresponding
Gaussian probability density function with the same desired moments, i.e., mean of zero
and variance of one. In this case, the Central Limit Theorem predicts that B should
approach a Gaussian distribution. From Figs. 2a-e, it can be seen that this occurs very
quickly with increasing N. For N = 1 (Fig. 2a), the first three moments are in excellent
agreement with the specified values ( B = 0.0 , B2 = 1.0 , and B3 = 0.0 ). With increasing
N, the fourth and higher moments approach the corresponding values for a Gaussian
distribution. For a Gaussian distribution with zero mean, the odd moments are zero and
the fourth moment is B4 = 3σ 4 = 3.0 .

Figs. 3a-e and 4a-e show calculations for non-zero skewness, S = 0.5 and 1.5,
respectively. (Skewness of 0.5 is typical for vertical velocities in the CBL.) These figures
show that the desired skewness is achieved quite accurately (to two decimal places) for
all values of N. For increasing N, smooth distributions are quickly obtained.

Figure 2a

Desired Skewness = 0.
Desired Mean     = 0.
Desired Variance = 1.0
N = 1
Number of final ran. num.= 1000000
Calculated Skewness of B   = 0.00225
Calculated Mean of B       = 0.00105
Calculated Variance of B   = 0.99829
Calculated 3rd moment of B = 0.00224
Calculated 4th moment of B = 2.24393
Calculated 5th moment of B = 0.00533

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 2b

Desired Skewness = 0.
Desired Mean     = 0.
Desired Variance = 1.0
N = 2
Number of final ran. num.= 1000000
Calculated Skewness of B   = -0.0035
Calculated Mean of B       = -0.0005
Calculated Variance of B   = 0.99948
Calculated 3rd moment of B = -0.0035
Calculated 4th moment of B = 2.62087
Calculated 5th moment of B = -0.0190

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 2c

Desired Skewness = 0.
Desired Mean     = 0.
Desired Variance = 1.0
N = 3
Number of final ran. num.= 1000000
Calculated Skewness of B   = -0.0042
Calculated Mean of B       = -0.0012
Calculated Variance of B   = 1.00090
Calculated 3rd moment of B = -0.0042
Calculated 4th moment of B = 2.75655
Calculated 5th moment of B = -0.0247

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 2d

Desired Skewness = 0.
Desired Mean     = 0.
Desired Variance = 1.0
N = 5
Number of final ran. num.= 1000000
Calculated Skewness of B   = -0.0028
Calculated Mean of B       = -0.0008
Calculated Variance of B   = 1.00209
Calculated 3rd moment of B = -0.0029
Calculated 4th moment of B = 2.86242
Calculated 5th moment of B = -0.0143

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 2e

Desired Skewness = 0.
Desired Mean     = 0.
Desired Variance = 1.0
N = 10
Number of final ran. num.= 1000000
Calculated Skewness of B   = -0.0021
Calculated Mean of B       = -0.0008
Calculated Variance of B   = 1.00205
Calculated 3rd moment of B = -0.0021
Calculated 4th moment of B = 2.93680
Calculated 5th moment of B = -0.0083

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 3a

Desired Skewness = 0.5
Desired Mean     = 0.
Desired Variance = 1.0
N =  1
Number of final ran. num.= 1000000
Calculated Skewness of B   = 0.50456
Calculated Mean of B       = 0.00118
Calculated Variance of B   = 0.99906
Calculated 3rd moment of B = 0.50385
Calculated 4th moment of B = 2.54383
Calculated 5th moment of B = 2.85461

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 3b

Desired Skewness = 0.5
Desired Mean     = 0.
Desired Variance = 1.0
N = 2
Number of final ran. num.= 1000000
Calculated Skewness of B   = 0.49820
Calculated Mean of B       = -0.0003
Calculated Variance of B   = 0.99951
Calculated 3rd moment of B = 0.49784
Calculated 4th moment of B = 2.91312
Calculated 5th moment of B = 3.98529

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 3c

Desired Skewness = 0.5
Desired Mean     = 0.
Desired Variance = 1.0
N = 3
Number of final ran. num.= 1000000
Calculated Skewness of B   = 0.49600
Calculated Mean of B       = -0.0016
Calculated Variance of B   = 0.99882
Calculated 3rd moment of B = 0.49513
Calculated 4th moment of B = 3.04044
Calculated 5th moment of B = 4.37807

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 3d

Desired Skewness = 0.5
Desired Mean     = 0.
Desired Variance = 1.0
N = 5
Number of final ran. num.= 1000000
Calculated Skewness of B   = 0.49894
Calculated Mean of B       = -0.0011
Calculated Variance of B   = 1.00133
Calculated 3rd moment of B = 0.49995
Calculated 4th moment of B = 3.15260
Calculated 5th moment of B = 4.71858

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 3e

Desired Skewness = 0.5
Desired Mean     = 0.
Desired Variance = 1.0
N = 10
Number of final ran. num.= 1000000
Calculated Skewness of B   = 0.49844
Calculated Mean of B       = -0.0010
Calculated Variance of B   = 1.00032
Calculated 3rd moment of B = 0.49868
Calculated 4th moment of B = 3.22919
Calculated 5th moment of B = 4.97588

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 4a

Desired Skewness = 1.5
Desired Mean     = 0.
Desired Variance = 1.0
N =  1
Number of final ran. num.= 1000000
Calculated Skewness of B   = 1.50333
Calculated Mean of B       = 0.00091
Calculated Variance of B   = 0.99837
Calculated 3rd moment of B = 1.49967
Calculated 4th moment of B = 4.90601
Calculated 5th moment of B = 12.7426

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 4b

Desired Skewness = 1.5
Desired Mean     = 0.
Desired Variance = 1.0
N =  2
Number of final ran. num.= 1000000
Calculated Skewness of B   = 1.49890
Calculated Mean of B       = 0.00006
Calculated Variance of B   = 0.99955
Calculated 3rd moment of B = 1.49791
Calculated 4th moment of B = 5.26957
Calculated 5th moment of B = 16.1751

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 4c

Desired Skewness = 1.5
Desired Mean     = 0.
Desired Variance = 1.0
N =  3
Number of final ran. num.= 1000000
Calculated Skewness of B   = 1.49754
Calculated Mean of B       = -0.0011
Calculated Variance of B   = 0.99859
Calculated 3rd moment of B = 1.49437
Calculated 4th moment of B = 5.38421
Calculated 5th moment of B = 17.2935

(Note: N = double-block random numbers, bi, per
final random number, B)

Figure 4d

Desired Skewness = 1.5
Desired Mean     = 0.
Desired Variance = 1.0
N =  5
Number of final ran. num.= 1000000
Calculated Skewness of B   = 1.50270
Calculated Mean of B       = -0.0005
Calculated Variance of B   = 1.00105
Calculated 3rd moment of B = 1.50508
Calculated 4th moment of B = 5.52842
Calculated 5th moment of B = 18.3748

(Note: N = double-block random numbers, bi, per
final random number, B)



23

Figure 4e

Desired Skewness = 1.5
Desired Mean     = 0.
Desired Variance = 1.0
N =  10
Number of final ran. num.= 1000000
Calculated Skewness of B   = 1.50213
Calculated Mean of B       = -0.0006
Calculated Variance of B   = 0.99922
Calculated 3rd moment of B = 1.50039
Calculated 4th moment of B = 5.61930
Calculated 5th moment of B = 19.4374

(Note: N = double-block random numbers, bi, per
final random number, B)

5. SUMMARY

The initial testing of probability distributions from combinations of double-block random
numbers has shown that this is a promising technique for generating skewed random
numbers. The method can generate random numbers with the desired first, second and
third moments. For calculations with zero skewness, the higher moments of a Gaussian
distribution were accurately reproduced when multiple double-block random numbers
were combined.

Ongoing and future work will use this skewed random number generator for the random
forcing term in a Langevin  model for skewed turbulence in the convective boundary
layer. The results will be compared with analytical results, laboratory and field
experiments, and other Langevin models.

APPENDIX

FORTRAN Double-block random number generator code

      program db

c____________________________________________________________________

c

c

c                      db (Double Block)

c

c

c____________________________________________________________________

c

c author:  John Nasstrom

c

c date:    April 1994

c

c purpose: Generate random numbers from double, overlapping "blocks"

c          of two uniform probability density functions using method

c          developed by Don Ermak

c

c_____________________________________________________________________

      implicit none

      integer ndb, nfinal, idb, ifinal, nbins, ibin

      parameter(nbins=48)

      integer*4 ranseed

      real mom2, skew, dbmom3, dbmean1, dbmean2, dbprob1,

     &     dbdelta1, dbdelta2, rana, ranb,

     &     dbran, sumdbran,

     &     sumdbran2, sumdbran3,

     &     finalran, sumfinalran, sumfinalran2, sumfinalran3,

     &     sumfinalran4, sumfinalran5,

     &     finalmean, finalmom2, finalmom3, finalmom4, finalmom5,

     &     finalskew,

     &     finalpdf(nbins), histbincount(nbins), histbinwidth,

     &     histbinlowval(nbins), histsigmas,

     &     a, b, c, terma, termb

      parameter(histsigmas=3.)

c--------------- Start execution ---------------

c----- Assign constants

      a = sqrt(5.)

      b = 2./9.

      c = 243./32.

      ranseed = 97531

c---- Read input parameters:

C     - Number of double-block random numbers (ndb) to sum for each

C       random number in final distribution

C     - Number of values of final random number to calculate (nfinal)

C     - Desired skewness (skew) and second moment (mom2) of final

C       distribution (mean is assumed to be zero)

      open(unit=1, name='dbin.dat', err=100, status='OLD')

      go to 120

100   continue

      stop  'Error opening input parameter file: dbin.dat'

120   continue

      read(1,*) ndb, nfinal, skew, mom2

      Close(unit=1)

c------ Open output file

      open(unit=2, name='dbout.dat', err=200, status='NEW')

      go to 220

200   continue

      stop ' Error opening output file: dbout.dat'

220   continue

      write(2,250) ndb, nfinal, skew, mom2

250   format(' ndb=',I8,'; nfinal=',I8, '; skew=', f7.3,

     &       '; mom2=', f7.3,';',/)

c-------- Calculate bins for histogram of final random numbers

      histbinwidth = (2.*histsigmas*sqrt(mom2))/float(nbins)

      do ibin = 1, nbins

         histbinlowval(ibin) = -histsigmas*sqrt(mom2)

     &                         + float(ibin-1)*histbinwidth

         histbincount(ibin) = 0.

      end do

c-------- Calculate parameters for double block (db) distribution---

      dbmom3 = sqrt(float(ndb)) * skew * (mom2**1.5)

      terma = b/mom2

      termb = sqrt(dbmom3**2 + c*mom2**3)

      dbmean1 = terma*(dbmom3-termb)

      dbmean2 = terma*(dbmom3+termb)

      dbprob1 = dbmean2/(2.*a*dbmean1*(dbmean1-dbmean2))

      dbdelta1 = -a*dbmean1

      dbdelta2 = a*dbmean2

      write(2,300) dbmom3,dbmean1,dbmean2,dbprob1,dbdelta1,dbdelta2

300   format( ' dbmom3=',   F16.10,  ';', /,

     &        ' dbmean1=',  F16.10,  ';', /,

     &        ' dbmean2=',  F16.10,  ';', /,

     &        ' dbprob1=',  F16.10,  ';', /,

     &        ' dbdelta1=', F16.10, ';', /,

     &        ' dbdelta2=', F16.10, ';', /

     & )

c------ Initialize sums for final random number moments ----

      sumfinalran = 0.

      sumfinalran2 = 0.

      sumfinalran3 = 0.

      sumfinalran4 = 0.

      sumfinalran5 = 0.

c--------------------------------------------------------------

c------ Loop for calculation of final random number -----------

c--------------------------------------------------------------

      Do ifinal = 1, nfinal

c----- Initialize values of final random number

         sumdbran = 0.

c--------------------------------------------------------------

c------ Loop for summation of double block random numbers -----

c------ in each final random number                       -----

c--------------------------------------------------------------

         Do idb = 1, ndb

         rana = ran(ranseed)

         ranb = ran(ranseed)

         if (rana .lt. (2.*dbdelta1*dbprob1) ) then

            dbran = dbmean1 + 2.*dbdelta1*ranb - dbdelta1

         else

            dbran = dbmean2 + 2.*dbdelta2*ranb - dbdelta2

         end if

c--------- add to sums for final random number

         sumdbran = sumdbran + dbran

         end do

c-----------end of loop for double-block random numbers-------------

c-------- Calculate final random number

         finalran = sumdbran / sqrt(float(ndb))

c-------- Increment count for appropriate bin of histogram of

c-------- final random number distribution

         do ibin = 1, nbins

            if (finalran.ge.histbinlowval(ibin) .and.

     &      finalran .lt. (histbinlowval(ibin)+histbinwidth) ) then

               histbincount(ibin) = histbincount(ibin) + 1.

               go to 450

            end if

         end do

450      continue

c-------- Add to sums for moments of final random numbers

         sumfinalran = sumfinalran + finalran

         sumfinalran2 = sumfinalran2 + finalran**2

         sumfinalran3 = sumfinalran3 + finalran**3

         sumfinalran4 = sumfinalran4 + finalran**4

         sumfinalran5 = sumfinalran5 + finalran**5

      end do

c---------end of loop for final random numbers----------------------

c------- Calculate moments of final random number distribution

      finalmean = sumfinalran / nfinal

      finalmom2 = sumfinalran2 / nfinal

      finalmom3 = sumfinalran3 / nfinal

      finalmom4 = sumfinalran4 / nfinal

      finalmom5 = sumfinalran5 / nfinal

      finalskew = finalmom3 / finalmom2**1.5

c------- Calculate final prodability density function (pdf)

      do ibin = 1, nbins

         finalpdf(ibin) = histbincount(ibin) / (nfinal*histbinwidth)

      end do

c------- Write out results in format readable by Mathematica

      write(2,1100) finalmean, finalmom2, finalmom3,

     &    finalmom4, finalmom5, finalskew

1100  format(' finalmean=', F16.10  ,';' ,/,

     &       ' finalmom2=',  F16.10 ,';' ,/,

     &       ' finalmom3=',  F16.10 ,';' ,/,

     &       ' finalmom4=',  F16.10 ,';' ,/,

     &       ' finalmom5=',  F16.10 ,';' ,/,

     &       ' finalskew=',  F16.10 ,';' ,/

     &       )

      write(2, 1200)

1200  format(' finalranhistogram = {' )

      do ibin = 1, nbins

         if(ibin.ne.nbins) then

            write(2, 1210) ibin, histbincount(ibin)

1210        format(' {',i3, ',', f10.0, '},')

         else

            write(2,1215) ibin, histbincount(ibin)

1215        format(' {',i3, ',', f10.0, '}')

         end if

      end do

      write(2, 1216)

1216  format(' } ;' )

      write(2, 1220)

1220  format(/,' finalranpdf = {' )

      do ibin = 1, nbins

         if(ibin.ne.nbins) then

            write(2, 1230) (histbinlowval(ibin)+histbinwidth/2.),

     &                  finalpdf(ibin)

1230        format(' {', f13.7, ',',  F16.10, '},')

         else

            write(2,1235) (histbinlowval(ibin)+histbinwidth/2.),

     &                  finalpdf(ibin)

1235        format(' {' , f13.7, ',',  F16.10, '} ')

         end if

      end do

      write(2, 1240)

1240  format(' } ;' )

      close(unit=2)

      stop 'normal'

      end







T
echnical Inform

ation D
epartm

ent  • Law
rence Liverm

ore N
ational Laboratory

U
niversity of C

alifornia • Liverm
ore, C

alifornia  94551


